Liouville Numbers

Manuel Eberl

December 14, 2021

Abstract

In this work, we define the concept of Liouville numbers as well as the standard construction to obtain Liouville numbers and we prove their most important properties: irrationality and transcendence.

This is historically interesting since Liouville numbers constructed in the standard way where the first numbers that were proven to be transcendental. The proof is very elementary and requires only standard arithmetic and the Mean Value Theorem for polynomials and the boundedness of polynomials on compact intervals.

Contents

1 Liouville Numbers

1.1 Preliminary lemmas

theory *Liouville-Numbers-Misc*
imports
Complex-Main
HOL−Computational-Algebra.Polynomial
begin

We will require these inequalities on factorials to show properties of the standard construction later.

lemma *fact-ineq*: $n \geq 1 \Longrightarrow \textit{fact}\ n + k \leq \textit{fact}\ (n + k)$
proof (*induction k*)
case (*Suc k*)
from *Suc* **have** $\textit{fact}\ n + \textit{Suc}\ k \leq \textit{fact}\ (n + k) + 1$ **by** *simp*
also from *Suc* **have** $\ldots \leq \textit{fact}\ (n + \textit{Suc}\ k)$ **by** *simp*
finally show *?case* **.**

qed *simp-all*

lemma *Ints-sum*:
  assumes $\bigwedge x.\ x \in A \Longrightarrow f\ x \in \mathbb{Z}$
  shows $sum\ f\ A \in \mathbb{Z}$
  by (*cases finite A*, *insert assms*, *induction A rule*: *finite-induct*)
    (*auto intro*!: *Ints-add*)

lemma *suminf-split-initial-segment'*:
  $summable\ (f :: nat \Rightarrow 'a::real\text{-}normed\text{-}vector) \Longrightarrow$
    $suminf\ f = (\sum n.\ f\ (n + k + 1)) + sum\ f\ \{..k\}$
 by (*subst suminf-split-initial-segment*[*of - Suc k*], *assumption*, *subst lessThan-Suc-atMost*)

    simp-all

lemma *Rats-eq-int-div-int'*: $(\mathbb{Q} :: real\ set) = \{of\text{-}int\ p\ /\ of\text{-}int\ q\ | p\ q.\ q > 0\}$
proof *safe*
  fix $x :: real$ **assume** $x \in \mathbb{Q}$
  then obtain *p q* **where** *pq*: $x = of\text{-}int\ p\ /\ of\text{-}int\ q\ q \neq 0$
    by (*subst* (*asm*) *Rats-eq-int-div-int*) *auto*
  show $\exists\, p\ q.\ x = real\text{-}of\text{-}int\ p\ /\ real\text{-}of\text{-}int\ q \wedge 0 < q$
  proof (*cases* $q > 0$)
    case *False*
    show *?thesis* **by** (*rule exI*[*of* - $-p$], *rule exI*[*of* - $-q$]) (*insert False pq*, *auto*)
  qed (*insert pq*, *force*)
qed *auto*

lemma *Rats-cases'*:
  assumes $(x :: real) \in \mathbb{Q}$
  obtains *p q* **where** $q > 0\ x = of\text{-}int\ p\ /\ of\text{-}int\ q$
  using *assms* **by** (*subst* (*asm*) *Rats-eq-int-div-int'*) *auto*

The following inequality gives a lower bound for the absolute value of an integer polynomial at a rational point that is not a root.

lemma *int-poly-rat-no-root-ge*:
  fixes $p :: real\ poly$ **and** $a\ b :: int$
  assumes $\bigwedge n.\ coeff\ p\ n \in \mathbb{Z}$
  assumes $b > 0\ poly\ p\ (a\ /\ b) \neq 0$
  defines $n \equiv degree\ p$
  shows $abs\ (poly\ p\ (a\ /\ b)) \geq 1\ /\ of\text{-}int\ b\ \hat{}\ n$
proof −
  let $?S = (\sum i{\leq}n.\ coeff\ p\ i * of\text{-}int\ a\ \hat{}\ i * (of\text{-}int\ b\ \hat{}\ (n - i)))$
  from ‹$b > 0$› **have** *eq*: $?S = of\text{-}int\ b\ \hat{}\ n * poly\ p\ (a\ /\ b)$
   by (*simp add*: *poly-altdef power-divide mult-ac n-def sum-distrib-left power-diff*)
  have $?S \in \mathbb{Z}$ **by** (*intro Ints-sum Ints-mult assms Ints-power*) *simp-all*
  moreover from *assms* **have** $?S \neq 0$ **by** (*subst eq*) *auto*
  ultimately have $abs\ ?S \geq 1$ **by** (*elim Ints-cases*) *simp*
  with *eq* ‹$b > 0$› **show** *?thesis* **by** (*simp add*: *field-simps abs-mult*)
qed

end

theory *Liouville-Numbers*
imports
  Complex-Main
  HOL−Computational-Algebra.Polynomial
  Liouville-Numbers-Misc
begin

A Liouville number is a real number that can be approximated well – but not perfectly – by a sequence of rational numbers. "Well", in this context, means that the error of the n-th rational in the sequence is bounded by the n-th power of its denominator.

Our approach will be the following:

- Liouville numbers cannot be rational.
- Any irrational algebraic number cannot be approximated in the Liouville sense
- Therefore, all Liouville numbers are transcendental.
- The standard construction fulfils all the properties of Liouville numbers.

1.2 Definition of Liouville numbers

The following definitions and proofs are largely adapted from those in the Wikipedia article on Liouville numbers. [1]

A Liouville number is a real number that can be approximated well – but not perfectly – by a sequence of rational numbers. The error of the n-th term $\frac{p_n}{q_n}$ is at most q_n^{-n}, where $p_n \in \mathbb{Z}$ and $q_n \in \mathbb{Z}_{\geq 2}$.
We will say that such a number can be approximated in the Liouville sense.

locale *liouville* =
  fixes *x* :: *real* **and** *p q* :: *nat* ⇒ *int*
  assumes *approx-int-pos*: *abs* (*x* − *p n* / *q n*) > *0*
    and *denom-gt-1*: *q n* > *1*
    and *approx-int*: *abs* (*x* − *p n* / *q n*) < *1* / *of-int* (*q n*) ^ *n*

First, we show that any Liouville number is irrational.

lemma (**in** *liouville*) *irrational*: $x \notin \mathbb{Q}$
proof
  assume $x \in \mathbb{Q}$

```
  then obtain c d :: int where d: x = of-int c / of-int d d > 0
    by (elim Rats-cases') simp
  define n where n = Suc (nat ⌈log 2 d⌉)
  note q-gt-1 = denom-gt-1[of n]

  have neq: c ∗ q n ≠ d ∗ p n
  proof
    assume c ∗ q n = d ∗ p n
    hence of-int (c ∗ q n) = of-int (d ∗ p n) by (simp only: )
    with approx-int-pos[of n] d q-gt-1 show False by (auto simp: field-simps)
  qed
  hence abs-pos: 0 < abs (c ∗ q n − d ∗ p n) by simp

  from q-gt-1 d have of-int d = 2 powr log 2 d by (subst powr-log-cancel) simp-all
  also from d have log 2 (of-int d) ≥ log 2 1 by (subst log-le-cancel-iff) simp-all
  hence 2 powr log 2 d ≤ 2 powr of-nat (nat ⌈log 2 d⌉)
    by (intro powr-mono) simp-all
  also have ... = of-int (2 ^ nat ⌈log 2 d⌉)
    by (subst powr-realpow) simp-all
  finally have d ≤ 2 ^ nat ⌈log 2 (of-int d)⌉
    by (subst (asm) of-int-le-iff)
  also have ... ∗ q n ≤ q n ^ Suc (nat ⌈log 2 (of-int d)⌉)
    by (subst power-Suc, subst mult.commute, intro mult-left-mono power-mono,
        insert q-gt-1) simp-all
  also from q-gt-1 have ... = q n ^ n by (simp add: n-def)
  finally have 1 / of-int (q n ^ n) ≤ 1 / real-of-int (d ∗ q n) using q-gt-1 d
    by (intro divide-left-mono Rings.mult-pos-pos of-int-pos, subst of-int-le-iff)
simp-all
  also have ... ≤ of-int (abs (c ∗ q n − d ∗ p n)) / real-of-int (d ∗ q n) using
q-gt-1 d abs-pos
    by (intro divide-right-mono) (linarith, simp)
  also have ... = abs (x − of-int (p n) / of-int (q n)) using q-gt-1 d(2)
    by (simp-all add: divide-simps d(1) mult-ac)
  finally show False using approx-int[of n] by simp
qed
```

Next, any irrational algebraic number cannot be approximated with rational numbers in the Liouville sense.

```
lemma liouville-irrational-algebraic:
  fixes x :: real
  assumes irrationsl: x ∉ ℚ and algebraic x
  obtains c :: real and n :: nat
    where c > 0 and ⋀(p::int) (q::int). q > 0 ⟹ abs (x − p / q) > c / of-int q
^ n
proof −
  from ‹algebraic x› obtain p where p: ⋀i. coeff p i ∈ ℤ p ≠ 0 poly p x = 0
    by (elim algebraicE) blast
  define n where n = degree p
```

```
— The derivative of p is bounded within {x − 1..x + 1}.
let ?f = λt. |poly (pderiv p) t|
define M where M = (SUP t∈{x−1..x+1}. ?f t)
define roots where roots = {x. poly p x = 0} − {x}

define A-set where A-set = {1, 1/M} ∪ {abs (x′ − x) |x′. x′ ∈ roots}
define A′ where A′ = Min A-set
define A  where A = A′ / 2
— We define A to be a positive real number that is less than 1::′a, 1 / M and
any distance from x to another root of p.

— Properties of M, our upper bound for the derivative of p:
have ∃ s∈{x−1..x+1}. ∀ y∈{x−1..x+1}. ?f s ≥ ?f y
  by (intro continuous-attains-sup) (auto intro!: continuous-intros)
hence bdd: bdd-above (?f ‘ {x−1..x+1}) by auto

have M-pos: M > 0
proof −
  from p have pderiv p ≠ 0 by (auto dest!: pderiv-iszero)
  hence finite {x. poly (pderiv p) x = 0} using poly-roots-finite by blast
  moreover have ¬finite {x−1..x+1} by (simp add: infinite-Icc)
  ultimately have ¬finite ({x−1..x+1} − {x. poly (pderiv p) x = 0}) by simp
  hence {x−1..x+1} − {x. poly (pderiv p) x = 0} ≠ {} by (intro notI) simp
  then obtain t where t: t ∈ {x−1..x+1} and poly (pderiv p) t ≠ 0 by blast
  hence 0 < ?f t by simp
  also from t and bdd have . . . ≤ M unfolding M-def by (rule cSUP-upper)
  finally show M > 0 .
qed

have M-sup: ?f t ≤ M if t ∈ {x−1..x+1} for t
proof −
  from that and bdd show ?f t ≤ M
    unfolding M-def by (rule cSUP-upper)
qed


— Properties of A:
from p poly-roots-finite[of p] have finite A-set
  unfolding A-set-def roots-def by simp
have x ∉ roots unfolding roots-def by auto
hence A′ > 0 using Min-gr-iff[OF ‹finite A-set›, folded A′-def, of 0]
  by (auto simp: A-set-def M-pos)
hence A-pos: A > 0 by (simp add: A-def)

from ‹A′ > 0› have A < A′ by (simp add: A-def)
moreover from Min-le[OF ‹finite A-set›, folded A′-def]
  have A′ ≤ 1 A′ ≤ 1/M ⋀x′. x′ ≠ x ⟹ poly p x′ = 0 ⟹ A′ ≤ abs (x′ − x)
  unfolding A-set-def roots-def by auto
ultimately have A-less: A < 1 A < 1/M ⋀x′. x′ ≠ x ⟹ poly p x′ = 0 ⟹ A
```

```
< abs (x' − x)
    by fastforce+


  — Finally: no rational number can approximate x “well enough”.
  have ∀(a::int) (b :: int). b > 0 ⟶ |x − of-int a / of-int b| > A / of-int b ^ n
  proof (intro allI impI, rule ccontr)
    fix a b :: int
    assume b: b > 0 and ¬(A / of-int b ^ n < |x − of-int a / of-int b|)
    hence ab: abs (x − of-int a / of-int b) ≤ A / of-int b ^ n by simp
    also from A-pos b have A / of-int b ^ n ≤ A / 1
      by (intro divide-left-mono) simp-all
    finally have ab': abs (x − a / b) ≤ A by simp
    also have … ≤ 1 using A-less by simp
    finally have ab'': a / b ∈ {x−1..x+1} by auto

    have no-root: poly p (a / b) ≠ 0
    proof
      assume poly p (a / b) = 0
      moreover from ‹x ∉ ℚ› have x ≠ a / b by auto
      ultimately have A < abs (a / b − x) using A-less(3) by simp
      with ab' show False by simp
    qed

    have ∃x0∈{x−1..x+1}. poly p (a / b) − poly p x = (a / b − x) ∗ poly (pderiv
p) x0
      using ab'' by (intro poly-MVT') (auto simp: min-def max-def)
    with p obtain x0 :: real where x0:
        x0 ∈ {x−1..x+1} poly p (a / b) = (a / b − x) ∗ poly (pderiv p) x0 by auto

    from x0(2) no-root have deriv-pos: poly (pderiv p) x0 ≠ 0 by auto

    from b p no-root have p-ab: abs (poly p (a / b)) ≥ 1 / of-int b ^ n
      unfolding n-def by (intro int-poly-rat-no-root-ge)

    note ab
    also from A-less b have A / of-int b ^ n < (1/M) / of-int b ^ n
      by (intro divide-strict-right-mono) simp-all
    also have … = (1 / b ^ n) / M by simp
    also from M-pos ab p-ab have … ≤ abs (poly p (a / b)) / M
      by (intro divide-right-mono) simp-all
    also from deriv-pos M-pos x0(1)
      have … ≤ abs (poly p (a / b)) / abs (poly (pderiv p) x0)
      by (intro divide-left-mono M-sup) simp-all
    also have |poly p (a / b)| = |a / b − x| ∗ |poly (pderiv p) x0|
      by (subst x0(2)) (simp add: abs-mult)
    with deriv-pos have |poly p (a / b)| / |poly (pderiv p) x0| = |x − a / b|
      by (subst nonzero-divide-eq-eq) simp-all
    finally show False by simp
```

```
  qed
  with A-pos show ?thesis using that[of A n] by blast
qed
```

Since Liouville numbers are irrational, but can be approximated well by rational numbers in the Liouville sense, they must be transcendental.

```
lemma (in liouville) transcendental: ¬algebraic x
proof
  assume algebraic x
  from liouville-irrational-algebraic[OF irrational this]
  obtain c n where cn:
    c > 0 ⋀p q. q > 0 ⟹ c / real-of-int q ^ n < |x − real-of-int p / real-of-int q|
    by auto

  define r where r = nat ⌈log 2 (1 / c)⌉
  define m where m = n + r
  from cn(1) have (1/c) = 2 powr log 2 (1/c) by (subst powr-log-cancel) simp-all
  also have log 2 (1/c) ≤ of-nat (nat ⌈log 2 (1/c)⌉) by linarith
  hence 2 powr log 2 (1/c) ≤ 2 powr … by (intro powr-mono) simp-all
  also have … = 2 ^ r unfolding r-def by (simp add: powr-realpow)
  finally have 1 / (2^r) ≤ c using cn(1) by (simp add: field-simps)

  have abs (x − p m / q m) < 1 / of-int (q m) ^ m by (rule approx-int)
  also have … = (1 / of-int (q m) ^ r) ∗ (1 / real-of-int (q m) ^ n)
    by (simp add: m-def power-add)
  also from denom-gt-1[of m] have 1 / real-of-int (q m) ^ r ≤ 1 / 2 ^ r
    by (intro divide-left-mono power-mono) simp-all
  also have … ≤ c by fact
  finally have abs (x − p m / q m) < c / of-int (q m) ^ n
    using denom-gt-1[of m] by − (simp-all add: divide-right-mono)
  with cn(2)[of q m p m] denom-gt-1[of m] show False by simp
qed
```

1.3 Standard construction for Liouville numbers

We now define the standard construction for Liouville numbers.

```
definition standard-liouville :: (nat ⇒ int) ⇒ int ⇒ real where
  standard-liouville p q = (∑k. p k / of-int q ^ fact (Suc k))

lemma standard-liouville-summable:
  fixes p :: nat ⇒ int and q :: int
  assumes q > 1 range p ⊆ {0..<q}
  shows   summable (λk. p k / of-int q ^ fact (Suc k))
proof (rule summable-comparison-test')
  from assms(1) show summable (λn. (1 / q) ^ n)
    by (intro summable-geometric) simp-all
next
  fix n :: nat
```

from *assms* **have** $p\ n \in \{0..<q\}$ **by** *blast*
with *assms*(*1*) **have** $\mathit{norm}\ (p\ n\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ \mathit{fact}\ (\mathit{Suc}\ n)) \le$
$q\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ n))$ **by** (*auto simp*: *field-simps*)
also from *assms*(*1*) **have** $\ldots = 1\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ n) - 1)$
by (*subst power-diff*) (*auto simp del*: *fact-Suc*)
also have $\mathit{Suc}\ n \le \mathit{fact}\ (\mathit{Suc}\ n)$ **by** (*rule fact-ge-self*)
with *assms*(*1*) **have** $1\ /\ \mathit{real\text{-}of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ n) - 1) \le 1\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ n$
by (*intro divide-left-mono power-increasing*)
(*auto simp del*: *fact-Suc simp add*: *algebra-simps*)
finally show $\mathit{norm}\ (p\ n\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ \mathit{fact}\ (\mathit{Suc}\ n)) \le (1\ /\ q)\ \hat{}\ n$
by (*simp add*: *power-divide*)
qed

lemma *standard-liouville-sums*:
assumes $q > 1$ $\mathit{range}\ p \subseteq \{0..<q\}$
shows $(\lambda k.\ p\ k\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ \mathit{fact}\ (\mathit{Suc}\ k))$ *sums standard-liouville p q*
using *standard-liouville-summable*[*OF assms*] **unfolding** *standard-liouville-def*
by (*simp add*: *summable-sums*)

Now we prove that the standard construction indeed yields Liouville numbers.

lemma *standard-liouville-is-liouville*:
assumes $q > 1$ $\mathit{range}\ p \subseteq \{0..<q\}$ *frequently* $(\lambda n.\ p\ n \ne 0)$ *sequentially*
defines $b \equiv \lambda n.\ q\ \hat{}\ \mathit{fact}\ (\mathit{Suc}\ n)$
defines $a \equiv \lambda n.\ (\sum k \le n.\ p\ k * q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ n) - \mathit{fact}\ (\mathit{Suc}\ k)))$
shows *liouville* (*standard-liouville p q*) *a b*
proof
fix $n :: \mathit{nat}$
from *assms*(*1*) **have** $1 < q\ \hat{}\ 1$ **by** *simp*
also from *assms*(*1*) **have** $\ldots \le b\ n$ **unfolding** *b-def*
by (*intro power-increasing*) (*simp-all del*: *fact-Suc*)
finally show $b\ n > 1$ **.**

note *summable* = *standard-liouville-summable*[*OF assms*(*1*,*2*)]
let $?S = \sum k.\ p\ (k + n + 1)\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ (k + n + 1)))$
let $?C = (q - 1)\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (n+2))$

have $a\ n\ /\ b\ n = (\sum k \le n.\ p\ k * (\mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (n+1) - \mathit{fact}\ (k+1))\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (n+1))))$
by (*simp add*: *a-def b-def sum-divide-distrib of-int-sum*)
also have $\ldots = (\sum k \le n.\ p\ k\ /\ \mathit{of\text{-}int}\ q\ \hat{}\ (\mathit{fact}\ (\mathit{Suc}\ k)))$
by (*intro sum.cong refl*, *subst inverse-divide* [*symmetric*], *subst power-diff* [*symmetric*])
(*insert assms*(*1*), *simp-all add*: *divide-simps fact-mono-nat del*: *fact-Suc*)
also have *standard-liouville p q* $- \ldots = ?S$ **unfolding** *standard-liouville-def*
by (*subst diff-eq-eq*) (*intro suminf-split-initial-segment′ summable*)
finally have *abs* (*standard-liouville p q* $-\ a\ n\ /\ b\ n) = \mathit{abs}\ ?S$ **by** (*simp only*:)
moreover from *assms* **have** $?S \ge 0$
by (*intro suminf-nonneg allI divide-nonneg-pos summable-ignore-initial-segment*

summable) *force*+
ultimately have *eq*: *abs* (*standard-liouville p q* − *a n* / *b n*) = *?S* **by** *simp*

also have *?S* ≤ (∑*k*. *?C* ∗ (*1* / *q*) ^ *k*)
proof (*intro suminf-le allI summable-ignore-initial-segment summable*)
from *assms* **show** *summable* (λ*k*. *?C* ∗ (*1* / *q*) ^ *k*)
by (*intro summable-mult summable-geometric*) *simp-all*
next
fix *k* :: *nat*
from *assms* **have** *p* (*k* + *n* + *1*) ≤ *q* − *1* **by** *force*
with ‹*q* > *1*› **have** *p* (*k* + *n* + *1*) / *of-int q* ^ *fact* (*Suc* (*k* + *n* + *1*)) ≤
(*q* − *1*) / *of-int q* ^ (*fact* (*Suc* (*k* + *n* + *1*)))
by (*intro divide-right-mono*) (*simp-all del*: *fact-Suc*)
also from ‹*q* > *1*› **have** . . . ≤ (*q* − *1*) / *of-int q* ^ (*fact* (*n*+*2*) + *k*)
using *fact-ineq*[*of n*+*2 k*]
by (*intro divide-left-mono power-increasing*) (*simp-all add*: *add-ac*)
also have . . . = *?C* ∗ (*1* / *q*) ^ *k*
by (*simp add*: *field-simps power-add del*: *fact-Suc*)
finally show *p* (*k* + *n* + *1*) / *of-int q* ^ *fact* (*Suc* (*k* + *n* + *1*)) ≤ . . . **.**
qed
also from *assms* **have** . . . = *?C* ∗ (∑*k*. (*1* / *q*) ^ *k*)
by (*intro suminf-mult summable-geometric*) *simp-all*
also from *assms* **have** (∑*k*. (*1* / *q*) ^ *k*) = *1* / (*1* − *1* / *q*)
by (*intro suminf-geometric*) *simp-all*
also from *assms*(*1*) **have** *?C* ∗ . . . = *of-int q* ^ *1* / *of-int q* ^ *fact* (*n* + *2*)
by (*simp add*: *field-simps*)
also from *assms*(*1*) **have** . . . ≤ *of-int q* ^ *fact* (*n* + *1*) / *of-int q* ^ *fact* (*n* + *2*)
by (*intro divide-right-mono power-increasing*) (*simp-all add*: *field-simps del*: *fact-Suc*)
also from *assms*(*1*) **have** . . . = *1* / (*of-int q* ^ (*fact* (*n* + *2*) − *fact* (*n* + *1*)))
by (*subst power-diff*) *simp-all*
also have *fact* (*n* + *2*) − *fact* (*n* + *1*) = (*n* + *1*) ∗ *fact* (*n* + *1*) **by** (*simp add*: *algebra-simps*)
also from *assms*(*1*) **have** *1* / (*of-int q* ^ . . .) < *1* / (*real-of-int q* ^ (*fact* (*n* + *1*) ∗ *n*))
by (*intro divide-strict-left-mono power-increasing mult-right-mono*) *simp-all*
also have . . . = *1* / *of-int* (*b n*) ^ *n*
by (*simp add*: *power-mult b-def power-divide del*: *fact-Suc*)
finally show |*standard-liouville p q* − *a n* / *b n*| < *1* / *of-int* (*b n*) ^ *n* **.**

from *assms*(*3*) **obtain** *k* **where** *k*: *k* ≥ *n* + *1 p k* ≠ *0*
by (*auto simp*: *frequently-def eventually-at-top-linorder*)
define *k*′ **where** *k*′ = *k* − *n* − *1*
from *assms k* **have** *p k* ≥ *0* **by** *force*
with *k assms* **have** *k*′: *p* (*k*′ + *n* + *1*) > *0* **unfolding** *k*′*-def* **by** *force*
with *assms*(*1*,*2*) **have** *?S* > *0*
by (*intro suminf-pos2*[*of - k*′] *summable-ignore-initial-segment summable*)
(*force intro*!: *divide-pos-pos divide-nonneg-pos*)+
with *eq* **show** |*standard-liouville p q* − *a n* / *b n*| > *0* **by** (*simp only*:)

qed

We can now show our main result: any standard Liouville number is transcendental.

theorem *transcendental-standard-liouville*:
  assumes $q > 1$ *range* $p \subseteq \{0..<q\}$ *frequently* $(\lambda k.\ p\ k \neq 0)$ *sequentially*
  shows $\neg$*algebraic* (*standard-liouville p q*)
proof –
  from *assms* **interpret**
    liouville standard-liouville p q
      $\lambda n.\ (\sum k{\le}n.\ p\ k * q\ \widehat{}\ (fact\ (Suc\ n) - fact\ (Suc\ k)))$
      $\lambda n.\ q\ \widehat{}\ fact\ (Suc\ n)$
    by (*rule standard-liouville-is-liouville*)
  from *transcendental* **show** *?thesis* **.**
qed

In particular: The the standard construction for constant sequences, such as the "classic" Liouville constant $\sum_{n=1}^{\infty} 10^{-n!} = 0.11000100\ldots$, are transcendental.

This shows that Liouville numbers exists and therefore gives a concrete and elementary proof that transcendental numbers exist.

corollary *transcendental-standard-standard-liouville*:
  $a \in \{0<..<b\} \Longrightarrow \neg$*algebraic* (*standard-liouville* (λ-. *int a*) (*int b*))
  by (*intro transcendental-standard-liouville*) *auto*

corollary *transcendental-liouville-constant*:
  $\neg$*algebraic* (*standard-liouville* (λ-. *1*) *10*)
  by (*intro transcendental-standard-liouville*) *auto*

end

References

[1] Wikipedia. Liouville number — Wikipedia, the free encyclopedia. https://en.wikipedia.org/w/index.php?title=Liouville_number&oldid=696910651, 2015. [Online; accessed 22-July-2004].